
GHC, le compilateur Haskell le plus utilisé, est sorti en version 8.0.2, vous pouvez consulter l'annonce de la sortie datée du 11 janvier 2017 ainsi que les notes de version


Il s'agit principalement d'une version de suivi qui corrige plus de deux cent bugs depuis la version 8.0 dont l'annonce de la sortie

avait été faite sur http://linuxfr.org. Donc à ce titre, il n'y a pas grand chose à raconter. Le futur, c'est GHC 8.2 qui devrait arriver aux environs d'avril si on en croit le planning. Celle-ci devrait apporter principalement des améliorations de performance de l'environnement d'exécution parallèle et du ramasse-miettes.


Comme ce journal frôle le journal bookmark et que je tient à mon karma, je vais vous présenter une fonctionnalité d'Haskell, que l'on retrouve dans de nombreux langages, j'ai nommé les ADT, où types de donnée algébriques


Il s'agit d'un croisement un peu douteux entre les struct, bien connus de nombreux langages de programmation, et les union, utilisés en C/C++ et qui sont une sorte d'enum. Le tout gonflé aux stéroïdes de la généricité, de la sécurité et du sucre syntaxique.


Remarque : ce journal a été rédigé dans l'espace de rédaction de dépêche, pour profiter de la collaboration pour corriger notamment de nombreuses coquilles. Merci aux participants, palm123, Anthony Jaguenaud et Lucas pour leur remarques bénéfiques.

ADT : un struct


Un ADT c'est un type. Commençons d'abord par la partie qui ressemble à des struct. Si je veux faire un point à 3 dimensions en Haskell, je ferai :


data Point = PointC Float Float Float deriving (Show)


Il s'agit de la déclaration du type Point. Ce type peut être construit grâce à un constructeur PointC qui prend 3 Float et renvoi un Point. Le type de ce constructeur est PointC :: Float -> Float -> Float -> Point, une fonction qui prend un Float, puis un autre Float, et encore un Float et renvoie un Point. Le nom du constructeur est libre, il aurait très bien pu être le même que le nom du type.


La clause deriving (Show) sert à générer automatiquement des fonctions pour l'affichage.


Un petit exemple de cas d'utilisation dans le shell interactif GHCI :


Prelude> PointC 1 2 3
PointC 1.0 2.0 3.0
Prelude> PointC 4 5 6
PointC 4.0 5.0 6.0


Le constructeur PointC peut aussi servir à déconstruire" les valeurs si il apparaît du coté gauche du signe = :


Prelude> PointC a b c = PointC 1 2 3
Prelude> a
1.0
Prelude> b
2.0
Prelude> c
3.0


Ceci est très pratique lors de la création de fonctions :


Prelude> getX (PointC x _ _) = x
Prelude> getY (PointC _ y _) = y
Prelude> getZ (PointC _ _ z) = z
Prelude> norm (PointC x y z) = sqrt (x * x + y * y + z * z)
Prelude> p = PointC 1 2 3
Prelude> getX p
1.0
Prelude> getY p
2.0
Prelude> getZ p
3.0
Prelude> norm p
3.7416575


Nous avons donc vu qu'un type en Haskell peut être vu comme un struct / objet dans d'autres langages, c'est à dire un agrégat de champs de type hétérogène. Si cela vous inquiète, on peut aussi donner des noms aux champs :


data Point = Point {x :: Float, y :: Float, z :: Float}


Mais ceci est une autre histoire.

ADT : un enum


Les "enum" dans de nombreux langages permettent de crée un type pouvant être représenté par plusieurs valeurs. L'exemple d'école :


data Couleur = Rouge | Vert | Bleu | Marron | Noir | Blanc deriving (Show)


Ici nous avons crée le type Couleur et nous lui avons associé 6 constructeurs différents. Observez bien le | entre les constructeurs, il représente l'alternative.


L'usage du nom "constructeur" ici est souvent troublante pour qui n'est pas habitué, dites vous simplement que Rouge est une fonction qui ne prend pas d'argument et renvoie une valeur de type Couleur, en ce sens, c'est un constructeur de Couleur.


On peut utiliser ces constructeurs pour créer des objets de type Couleur:


Prelude> Rouge
Rouge
Prelude> Bleu
Bleu


On peut aussi réaliser différentes opérations dessus grâce à de la déconstruction comme vu précédemment. Dans la fonction réaction qui suit, je liste les différents cas de couleur, le _ servant de joker. En Haskell, on peut définir des fonctions en listant les cas :


réaction Rouge = "Cool"
réaction Bleu = "Cool aussi, mais je préfère le rouge"
réaction Vert = "Bof Bof"
réaction Noir = "Moche"
-- cas générique
réaction _ = "Je n'aime pas les autres couleurs"


Et l'usage dans l'interpréteur nous donne :


Prelude> réaction Rouge
"Cool"
Prelude> réaction Blanc
"Je n'aime pas les autres couleurs"


Nous avons vu comment réaliser en Haskell l'équivalent des "enum" que l'on peut trouver dans d'autres langages.

ADT : enum avancés, ou union


Le C et le C++ proposent un mécanisme d'union, où un type peut contenir au choix plusieurs sous-types. Par exemple :


union Forme
{
     struct {
          float cote;
     } carre;

     struct {
          float coteA;
          float coteB;
     } rectangle;
};


Je ne reviendrai pas sur son usage en C, sachez seulement que le type Forme peut contenir soit un float cote, soit deux float coteA et coteB. L'usage des deux simultanément est indéfini et on ajoute souvent à la structure de donnée un marqueur pour informer l'utilisateur du type de la donnée réellement stockée.


En haskell, ce type peut être facilement représenté par un ADT combinant struct (ou type "produit") et enum (ou type "somme") :


data Forme = Carré Float | Rectangle Float Float deriving (Show)


Ici nous avons un type Forme qui contient deux constructeurs :

    - Carré, qui associe un Float à une Forme

    - Rectangle, qui associe deux Float à une Forme.


Contrairement aux langages qui supportent les enum où les unions, on remarque notamment que n'importe quel type complexe peut apparaître dans les différents cas de l'énumération.


Les outils décrit précédemment, de construction et de déconstruction par analyse de cas fonctionnent également. Ainsi on peut crée des Forme :


Prelude> Carré 10
Carré 10.0
Prelude> Rectangle 20 30
Rectangle 20.0 30.0


Et on peut faire des fonctions qui vont traiter notre type par analyse de cas :


surface (Carré c) = c * c
surface (Rectangle a b) = a * b



Ici la fonction surface déconstruit un Carré et renvoie sa surface. Si la déconstruction n'est pas possible (car c'est un Rectangle), alors la fonction passe au cas suivant). À l'usage :


Prelude> surface (Carré 10)
100.0
Prelude> surface (Rectangle 5 3)
15.0


Plusieurs remarques :



	
Le compilateur nous protège et ce de plusieurs manières :



	Si j'avais oublié de gérer les cas Rectangle, le compilateur m'aurait prévenu.

	Contrairement aux unions en C/C++, on ne peut pas confondre un Rectangle et un Carre, c'est de nombreuses heures de recherche d'erreurs qui disparaissent soudainement.





	La syntaxe et l'usage sont succincts, c'est agréable à mon goût. La même chose est possible dans de nombreux langages, par exemple en C++, grâce à l'utilisation de "variant" mais l'usage est lourd. Comparez le programme entier en Haskell à la version C++ :



data Forme = Carré Float | Rectangle Float Float

surface (Carré c) = c * c
surface (Rectangle a b) = a * b

main = do
   let carré = Carré 10
       rectangle = Rectangle 5 3

   print (surface carré)
   print (surface rectangle)


La version C++ équivalente suivante utilise boost::variant, en c++17 nous utiliserons std::variant :


#include <iostream>
#include <boost/variant.hpp>

struct Carre
{
    float c;
};

struct Rectangle
{
    float a;
    float b;
};

using Forme = boost::variant<Carre, Rectangle>;

class surface
{
public:
    float operator()(const Carre &carre) const
    {
        return carre.c * carre.c;
    }

    float operator()(const Rectangle &rectangle) const
    {
        return rectangle.a * rectangle.b;
    }
};

int main()
{
    Forme carre = Carre{10};

    Forme rectangle = Rectangle{5, 3};

    // affiche 100
    std::cout << boost::apply_visitor(surface(), carre) << std::endl;
    // affiche 15
    std::cout << boost::apply_visitor(surface(), rectangle) << std::endl;
}


Ce code passe par la définition du type en trois étapes : définition des sous types Carre et Rectangle et définition du type Forme comme un variant, un choix, entre les deux précédents types.


La classe surface est ici un visiteur qui propose une surcharge de l'opérateur float operator(const T &t) pour chaque sous type T que peut contenir notre variant.


La fonction boost::apply_visitor est chargée d'appeler la bonne surcharge de l’opérateur operator() de surface en fonction du contenu du variant passé en second paramètre.

ADT : exemple plus poussé


Alors pourquoi je vous raconte tout cela. En premier lieu, j'aime bien. En second lieu, je me dis que cela peut vous intéresser à Haskell ou au moins vous sensibiliser à l'existence de ce type d'outil et peut-être que vous les utiliserez dans vos projets futurs. Par exemple, dans mon travail quotidien, je fais du C++, mais Haskell m'a beaucoup influencé et j'utilise tous les jours des boost::variant. Mon opinion là-dessus c'est que même si la syntaxe en C++ est verbeuse à souhait, cela sauve de certaines situations. Au final je pense que le code est plus robuste.


Pour finir, je vais vous donner un exemple d'un problème que je rencontre souvent dans des API et qui serait, à mon sens, mieux traité avec des ADT. C'est le cas traditionnel des valeurs sentinelles. Je fais un peu concurrence au journal de cette semaine sur la prévention de bug en Ocaml grâce à un typage plus strict. Là où ce journal s'intéressait à la définition d'un type synonyme mais incompatible, je m'intéresse à la définition d'un type totalement diffèrent permettant de représenter un contexte différent et ainsi de supprimer les cas impossibles et de rendre plus robustes les cas possibles.

Introduction


Pour appuyer mon propos, intéressons-nous à une libraire C++ que j'utilise tous les jours, OpenEXR, qui permet, entre autre, de lire et d'écrire des images au format EXR très utilisé dans l'industrie de l'image. La page github d'OpenEXR.


Cette librairie propose entre autre la lecture et l'écriture de fichiers via plusieurs threads, ce qui est une fonctionnalité très pratique quand l'écriture de plusieurs Go d'images en séquentiel est le facteur limitant sur des machines à 24 coeurs.


Le point de l'API qui nous intéresse est le suivant, les fonctions setGlobalThreadCount et globalThreadCount :


void    setGlobalThreadCount (int count);
int globalThreadCount();


Alors, si on lit la documentation, on peut voir que count dans setGlobalThreadCount sert à définir le nombre de thread utilisés globalement pour réaliser les écritures.


En cherchant un peu, on tombe sur ce commentaire :




	The functions in this file query and control the total number
of worker threads, which will be created globally for the whole
library.  Regardless of how many input or output files are
opened simultaneously, the library will use at most this number
of worker threads to perform all work.  The default number of
global worker threads is zero (i.e. single-threaded operation;
everything happens in the thread that calls the library).






Traduction à l'arrache :



La fonction setGlobalThreadCount controle le nombre total de threads […] la bibliothèque utilisera au maximum ce nombre de threads. Le nombre par défaut est zéro, ce qui signifie que les opérations ne seront pas parallélisées.




On tombe aussi sur des discussion github intéressantes, dont je ne trouve plus le lien, désolé, traitant du moyen de fournir à setGlobalThreadCount une valeur qui correspond au nombre de thread système optimal (sans avoir à trouver celui-ci, on peut imaginer qu'il puisse changer à la volée en fonction de la charge de la machine, où dans un environnement virtualisé, en fonction des besoins), et les débats tournaient autour du fait de mettre -1 comme nombre de thread pour ce cas de figure. Ce n'est pas implémenté à ma connaissance dans openEXR, mais supposons que cela le soit.


Donc en gros, voici le comportement que nous pouvons imaginer pour setGlobalThreadCount(n) :



	Si n > 0, alors c'est le nombre de thread utilisé globalement

	Si n = 0, alors il n'y aura pas de multi threading

	Si n = -1, alors on utilise le nombre de thread machine

	Si n = -12, autre cas particulier que nous pourrions imaginer.


Le problème


Premier problème, en tant qu'utilisateur, je n'avais pas conscience de l'existence des cas 0, -1 et -12 sans lire le code source et la documentation d'OpenEXR.


Second problème, on va se planter, LARGEMENT, en beauté. Qui ? Les développeurs d'OpenEXR sans doute, et moi en utilisant leur API. Comment je le sais ? Parce que je me suis planté.


Où pouvons-nous nous planter ? Partout où le nombre global de thread est utilisé. Si le cas particulier 0, -1 et -12 n'est pas géré explicitement, et bien c'est un bug. Cela peut faire des choses marrantes, comme par exemple créer 0 thread de travail, et répartir le travail entre eux, ce qui donne un blocage de l'application.


Troisième problème, le futur. Même si c'est bien géré actuellement, que se passe-t-il demain lorsque quelqu'un va écrire une nouvelle fonction basée sur cette valeur sans connaître l'existence des cas particuliers qui existent ? Et bien cela va marcher, jusqu'à ce que quelqu'un utilise un cas particulier non traité, et là, pan. Ou si quelqu'un ajoute un nouveau cas particulier et ne le gère pas à tous les endroits nécessaires ?


On peut aussi se planter en passant une mauvaise constante par erreur. Imaginons qu'il existe dans le même espace de nom, une constante nommée "NoThreading", mais utilisée par une autre librairie, et ayant pour valeur magique un entier. Si celui-ci est négatif, c'est le drame, le comportement du programme est largement indéfini, au mieux c'est une erreur à l'exécution, au pire ?. Si celui-ci est positif, il faut espérer qu'il ne soit pas trop gros, car je n'aimerais pas créer 100000 threads sur ma machine de production, l'OS ne tiendrait pas.


Ce type de bug potentiel est la raison qui fait que la montée de version sur un gros projet logiciel est difficile du fait de la peur des régressions. Et même le meilleur système de test unitaire ne peut rien garantir à ce sujet.


Je passe aussi sur le problème de documentation et de lecture de code avec l'utilisation de constantes magiques en paramètre de fonction. setGlobalThreadCount(-123) n'est pas très informatif. Alors oui, cela se règle avec des définitions de constante, mais on peut encore se tromper, en définissant la constante à une mauvaise valeur, et rien ne force le développeur à utiliser la constante magique.


Ce problème est présent de partout, dans toutes les bibliothèques que nous utilisons, dans tous les langages que nous utilisons. Ceux-ci proposent des valeurs sentinelles. Python avec la méthode find des chaînes de caractères, qui renvoie -1 si la chaîne n'est pas trouvés (Il y a la version avec exception, c'est moins pire). C++ avec la fonction std::find qui retourne un itérateur vide en cas d'échec et rien qui ne vous force à tester cela.

La solution


La solution passe par la définition d'un type représentant le problème plus finement. Dans le cas d'OpenEXR, et si celui-ci était écrit en Haskell, nous pourrions avoir un type :


-- Word est en entier non signé
data PolitiqueThread = NombreFixé Word | NombreMaximumHardware | PasDeThreading | CasParticulier


Ainsi on pourrait appeller la fonction setGlobalThreadCount de différentes façon :


setGlobalThreadCount (NombreFixé 8)

setGlobalThreadCount NombreMaximumHardware

setGlobalThreadCount PasDeThreading

setGlobalThreadCount CasParticulier


Nous réglons en premier lieu le problème de documentation lors de l'appel. En tant qu'utilisateur, je suis forcé de voir que ce n'est pas juste un entier, et d'au moins voir la documentation avec la liste des cas, qui est automatiquement à jour. Le code est lisible et il est explicite que cette valeur n'est pas anodine.


Nous réglons aussi le problème lors de l'usage. On ne peut plus utiliser les valeurs -1 et -12 et 0 par erreur en considérant qu'il s'agit d'un nombre de thread et non pas d'un cas particulier, car le langage nous force à déconstruire et à gérer les différents cas de déconstruction. Observez comment 0, -1 et -12 n'apparaissent pas :


threadCount <- getGlobalThreadCount
case threadCount of
   NombreFixé n -> "nombre fixé à " ++ show n
   NombreMaximumHardware -> "Fait chauffer la ferme de calcul"
   PasDeThreading -> "Plutôt tranquille"
   CasParticulier -> "Celui-ci je ne l'aime pas"


Nous réglons aussi le problème de l'évolution future et de l'ajout de nouveau cas particulier, puisque le compilateur vas râler aux endroits où tous les cas ne sont pas gérés.


Le problème de passer une valeur qui n'a pas de sens par défaut n'existe plus non plus. Le type PolitiqueThread est incompatible avec un autre type. La seul erreur possible reste de passer un nombre qui n'a pas de sens à NombreFixé. Soit un nombre négatif, soit un nombre trop grand qui ferait exploser le système.


Je n'ai pas de solution parfaite à ce dernier problème. On peut en premier lieu cacher le constructeur NombreFixé et le remplacer par une fonction type :


nombreFixé n
 | n > 0 && n < maximumThread = NombreFixé (fromIntegral n)
 | otherwise = erreurRuntime ("n est trop bizarre: " ++ show n)

-- fromIntegral sert à convertir n qui est un entier signé vers un `Word`.


Cette solution limite la casse. Il y en d'autres. On pourrait par exemple utiliser de l'analyse statique de code en imposant des contraintes sur nombreFixé. Liquid Haskell sait faire cela, mais dans un contexte limité.

Conclusion


En profitant de la sortie de GHC 8.0.2, j'ai essayé de vous sensibiliser un peu plus au problème des valeurs sentinelles. À mon sens, ce problème est grave car il en découle du code peu lisible, peu "naturellement" documenté, peu robuste à l'évolution et peu robuste à l'utilisation normale par un développeur qui ne connaît pas par cœur les détails de l'API qu'il utilise. Une solution est l'emploi d'ADT, ceux-ci sont disponibles dans de nombreux langages, comme Haskell, Caml, Rust, Scala, Swift, … Et peuvent plus ou moins facilement être remplacés par des structures équivalentes à la simplicité près, comme avec les variant en C++.


Ce que je vous ai montré n'est qu'une partie de ce qui est possible avec les ADTs, et un lecteur motivé pourra commencer sa lecture sur la section de wikipedia consacrée aux ADT généralisés

